BiCMOS BJT Noise Voltage Test

```
*
*       ________
*    _|_       _|_
*    ///     /     \
*            \/‾\/ I1
*            /\_/\
*            \___/   1KOhm = 4nV/rt_Hz
*       _______|
*      |      _|
*      |____|'
*     VB    |`->
*             _|_ VE
*            /VIN\
*            \___/
*             _|_
*             ///
*
*

vin                  VE   0   DC    0.0 ac 1.0u
I1                   0    VB 1uA
q1                   VB   VB  VE    BiCMOS_NPN


.model               BiCMOS_NPN  npn kf=.3e-16 af=.75 bf=200 rb=1000
.control
setplot              new

let                 "V1u"    =     0*vector(41)
let                 "V10u"   =     0*vector(41)
let                 "V100u"  =     0*vector(41)
let                 "V1000u" =     0*vector(41)

op
noise               v(vb) vin dec 10 10 100k 1
destroy
let                 unknown.V1u = sqrt(v(onoise_spectrum))

alter               I1      dc  = 10u
op
noise               v(vb) vin dec 10 10 100k 1
destroy
let                 unknown.V10u = sqrt(v(onoise_spectrum))

alter               I1      dc  = 100u
op
noise               v(vb) vin dec 10 10 100k 1
destroy
let                 unknown.V100u = sqrt(v(onoise_spectrum))

alter               I1      dc  = 1000u
op
noise               v(vb) vin dec 10 10 100k 1
destroy
let                 unknown.V1000u = sqrt(v(onoise_spectrum))


set pensize =         2
plot unknown.V1u unknown.V10u unknown.V100u unknown.V1000u vs frequency loglog title
BiCMOS_NPN

echo                "    ... done."
.endcontrol
.end
```

===============END==================================

`.model                  BiCMOS_NPN  npn kf=.3e-16 af=.75 bf=200 rb=1000`

In this case the BiCmos NPN Model is showing what to expect in a BiCMOS world. The base resistance has been set to 1kOhms such that at currents above 30uA, the flat band noise should be the same as a 1KOhm resistor (4nV_per_rt_Hz). At the higher currents, the 1/f noise corner should be expected to increase in frequency.



Spice likes to think in terms of power. Therefore a square root function needs to be applied to the output noise. In the output noise is onoise_spectrum. The refered to input noise is the should equal the output noise divided by the gain.